
15. PROGRAMLAR: Doğrusal denklem sistemi çözümü-CholeskyBant11 
 

Mühendislikte karşılaşılan problemlerin çoğunda A x=b denklem sisteminin Anxn katsayılar 
matrisi simetrik (AT=A), pozitif tanımlı (xT A x>0) ve bant şeklindedir. 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
Bilgisayar belleğinden yer kazanmak amacıyla A nın sadece bant içindeki satırları A(nxiBant) 
tek boyutlu matrisinde depolanır: 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
Bnxm nin kolonlarının depolanması:  [b11 b21 …bn1| b12 b22 …bn2| …| b1m b2m …bnm] 
 
Yukarıdaki şekilde x bant içinde ve sıfırdan farklı olduğu varsayılan terimleri göstermektedir. 
iBant bant genişliğidir, diyagonal ve diyagonali izleyen sıfırdan farklı terim sayısıdır: 
1≤iBant≤n dir.  
 

Bu tür denklem sisteminin çözümü için CholeskyBant1 alt programı kullanılabilir. Ekte 
QBASIC kodu verilen programda Anxn simetrik matrisinin bant içindeki satırları A(n x iBant) 
tek boyutlu matrisinde, bnxm nin kolonları b(nxm) tek boyutlu matriste depolanmış olmalıdır.  
 

CholeskyBant1 alt programı x bilinmeyenler matrisini ve ayrıca d, k gibi iki değer hesaplar. 
A nın determinantı 
 

det A=d.2k 
 

ile hesaplanabilir. d=0 olması durumunda A tekildir veya pozitif tanımlı değildir, çözüm 
yoktur. 
 

Hesaplanan x çözümü b matrisinde depolanır. Bu nedenle x matrisi için programda boyut 
açılmamıştır. Hesap sonrası A ve b nin ilk değerleri kaybolur. A=UTU çarpanlarından U üst 
üçgeni(bant formunda) A matrisinde depolanır. 
 
b=I birim matris ve m=n olarak verilirse çözüm A-1 olur. 

                                                 
1 Teori ve sayısal örnekler için bak: bölüm 6  

1. satır 2. satır n. satır 

A nın satırlarının Tek 
boyutlu matriste 
depolanması 

1. kolon 2. kolon m. kolon 

Simetrik 

Örnek: 

Katsayılar matrisi A nın programda depolanışı: 
 
 
 
 
 
 
Karşı taraf matrisi b nın programda depolanışı: 
 
 
 
 
b�[10 1 7 -11 3 8 14 10 -10 7 -11 3 8 14] 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
Çözüm: 

x

Simetrik 

iBant=4 

A�[10 2 -3 -1 9 2 -1 3 12 3 0 5 8 1 -2  -2 6 1 2 0 9 3 0 0 10 0 0 0] 

1. satır 2. satır n. satır 

1. kolon 1. kolon 

3. satır 

CholeskyBant1 sonucu 

'--------------- Ana program CholeskyBant1------------ 
' Ax=b denklem sistemi çözümü 
' A simetrik pozitif tanımlı bant matris 
' Çağrılan programlar: CholeskyBant1, AsagiyukariHesap 
'------------------------------------------------- 
DATA 7: 'Denklem sayısı n 
DATA 4: 'Band genişliği iBant 
DATA 2: 'karşı taraf sayısı m 
'A matrisi: 
DATA 10,2,-3,-1 
DATA 9,2,-1,3 
DATA 12,3,0,5 
DATA 8,1,-2,-2 
DATA 6,1,2,0 
DATA 9,3,0,0 
DATA 10,0,0,0 
'b matrisi: 
DATA 10,1,7,-11,3,8,14 
DATA 10,-10,7,-11,3,8,14 
 
DEFINT I-N 
DEFDBL A-H, O-Z 
 
DECLARE SUB CholeskyBant1 (n, iBant, a(), d, k) 
DECLARE SUB AsagiYukariHesap (n, iBant, a(), m, b()) 
    
    READ n, iBant, m 
    CLS 
    DIM a(n * iBant) ' A  için yer ayır 
    DIM b(n * m): ' b için yer ayır 
 
'A nın bant içindeki terimlerini oku 
    FOR I = 1 TO n * iBant 
     READ a(I) 
    NEXT I 
'B yi oku 
    FOR I = 1 TO n * m 
     READ b(I) 
    NEXT I 
 
    CALL CholeskyBant1(n, iBant, a(), d, k) 
 
    IF d = 0 THEN PRINT "Matris pozitif tanımlı değil !" : END 
 
    CALL AsagiYukariHesap(n, iBant, a(), m, b()) 
 
    PRINT "Determinant ve kondisyon hesabı için katsayılar:" 
    PRINT "d="; d, "k="; k 
    PRINT 
    PRINT "Denklem sisteminin çözümü (CholeskyBant1):" 
    FOR I = 1 TO m 
     FOR j = 1 TO n 
      PRINT b((I - 1) * n + j), 
     NEXT j 
     PRINT : PRINT 
    NEXT I 
 
END 'CholeskyBant1 ana 
 

Ana program 

A nın üst üçgeninin 
bant içindeki satırları 

B nin kolonları 

A ve B matrisleri için bellekte yer ayrılıyor 

A nın bant içindeki satırları DATA satırlarından okunuyor 

B nin kolonları DATA satırlarından okunuyor 

CholeskyBant1 alt programı 
çağrılıyor 

Sonuçlar yazdırılıyor 

Denklem sayısı, bant genişliği ve karşı taraf sayısı okunuyor 

Denklem sayısı, bant genişliği ve karşı taraf sayısı verileri 

AsagiYukariHesap alt programı çağrılıyor 

SUB CholeskyBant1 (n, iBant, a(), d, k) 
'------------------------------------------------------------------- 
' Doğrusal denklem sistemi çözümü 
' Dr. Ahmet TOPÇU, Osmangazi Üniversitesi, ESKİŞEHİR, 1994 
' A(n,n)*x(n,m)=b(n,m) denklem sisteminin x(n,m) vektorü hesaplanır. 
' CholeskiBant1 A matrisini üçgen çarpanlarına ayırır. Denklem sisteminin 
' çözümü için CholeskiBant1 in hemen arkasından AsagiYukariHesap alt 
' programının da çağrılması gerekir. 
' n denklem sayısı, iBant yarı bant genişliğidir. 
' A katsayılar matrisi simetrik ve pozitif tanımlı olmalıdır. 
' A üst üçgeni A(n*iBant) tek boyutlu alanında sıra ile depolanmış olmalıdır. 
' b nın kolonları b(n*m) matrisinde sıra ile depolanmış olmalıdır. 
' x(n,m) çözümü b(n*m) matrisinde depolanır. 
' A nın determinantı Det a=d*2^k ile hesaplanabilir. 
' Program d=0 değeri ile dönerse, a matrisi tekildir veya pozitif 
' tanımlı değildir ve CholeskyBant1 çözüm vermez. 
' Bu programın FORTRAN kodu "DANKERT,j., Numerische Methoden der Mechanik, 
' Springer, 1977" adlı kaynaktan alınmıştır. 
'------------------------------------------------------------------- 
' Machep 
  Eps = 1 
  DO 
   Eps = Eps / 2 
   s = 1 + Eps 
  LOOP UNTIL s <= 1 
  Eps = 2 * Eps 
 
'Zero: sıfır varsayılacak sayı 
    Zero = 0 
    FOR I = 1 TO n * iBant STEP iBant 
     IF ABS(a(I)) > Zero THEN Zero = ABS(a(I)) 
    NEXT I 
    Zero = Zero * Eps 
'Çarpanlara ayır 
    IF n = 1 OR iBant = 1 THEN 
     FOR I = 1 TO n 
      IF a(I) <= Zero GOTO 100 
      a(I) = SQR(a(I)) 
     NEXT I 
     GOTO 400 
    ELSE 
     IF a(1) <= Zero GOTO 100 
     a(1) = SQR(a(1)) 
      FOR I = 2 TO iBant 
       a(I) = a(I) / a(1) 
      NEXT I 
     END IF 
    
    iBw1 = iBant - 1 
    irue = iBw1 * iBw1 - 1 
    IF iBant > 2 THEN 
     iE = iBant 
     FOR I = 2 TO iBw1 
      s = a(iE + 1) 
      FOR j = I TO iE STEP iBw1 
       s = s - a(j) ^ 2 
      NEXT j 
      IF s <= Zero THEN GOTO 100 
      s = SQR(s) 
      a(iE + 1) = s 
      s = 1 / s 
      iEe = iE + 1 
      j = I + 1 
      kkk = I 
 
8     IF j > iBant THEN 
       j = j + iBw1 
       kkk = kkk + iBw1 
      END IF 
      IF j <= iE THEN 
       iEe = iEe + 1 
       ss = a(iEe) 
       kk = kkk 
       FOR k = j TO iE STEP iBw1 
        ss = ss - a(k) * a(kk) 
        kk = kk + iBw1 
       NEXT k 
       a(iEe) = ss * s 
       j = j + 1 
       GOTO 8 
      END IF 
      iE = iE + iBant 
      a(iE) = a(iE) * s 
     NEXT I 
    END IF 
 

    iE = iBw1 * iBant 
    FOR I = iBant TO n 
     s = a(iE + 1) 
     ia = iE - irue 
     FOR j = ia TO iE STEP iBw1 
      s = s - a(j) ^ 2 
     NEXT j 
     IF s <= Zero GOTO 100 
     s = SQR(s) 
     a(iE + 1) = s 
     s = 1 / s 
     IF iBant <> 2 THEN 
      iaa = ia + iBant 
      iEe = iE + 1 
      FOR j = iaa TO iE STEP iBant 
       ia = ia + iBw1 
       iEe = iEe + 1 
       ss = a(iEe) 
       kk = j 
       FOR k = ia TO iE STEP iBw1 
        ss = ss - a(k) * a(kk) 
        kk = kk + iBw1 
       NEXT k 
       a(iEe) = ss * s 
      NEXT j 
     END IF 
     iE = iE + iBant 
     a(iE) = a(iE) * s 
    NEXT I 
 
' determinant (WILKINSON) 
400 d = 1: k = 0 
    FOR I = 1 TO n 
     j = (I - 1) * iBant + 1 
     d = d * a(j) * a(j) 
500  k = k + 4 
     IF ABS(d) >= 1 THEN d = d * .0625: GOTO 500 
600  k = k - 4 
     IF ABS(d) < .0625 THEN d = d * 16: GOTO 600 
    NEXT I 
    EXIT SUB 
 
100 'Matris tekil veya pozitif tanımlı değil 
     d = 0: k = 0 
END SUB  
 

 

CholeskyBant1 alt programı 
 

A=UTU 
 

Çarpanlarını hesaplar U üst 
üçgenini  A da depolar 

SUB AsagiYukariHesap (n, iBant, a(), m, b()) 
'---------------------------------------------------- 
' CholeskyBant1 alt programından hemen sonra çağrılır. 
' Üçgen çarpanlarına ayrılmıŸ katsayılar matrisini 
' kullanarak bilinmeyenler matrisini hesaplar 
'---------------------------------------------------- 
   iBw1 = iBant - 1 
   nm = n * m 
   nibw = n * iBant 
   n1 = n - 1 
   IF iBant = 1 THEN 
    FOR I = 1 TO m 
     ia = (I - 1) * n 
     FOR j = 1 TO n 
      b(ia + j) = b(ia + j) / a(j) ^ 2 
     NEXT j 
    NEXT I 
    EXIT SUB 
   END IF 
 
   FOR I = 1 TO nm STEP n 
    b(I) = b(I) / a(1) 
    ii = I - 1 
    jja = 0 
    ia = iBant + 1 
    FOR j = 2 TO n 
     ij = ii + j 
     s = b(ij) 
     jog = ij - 1 
     IF j >= iBant THEN 
     jug = ij - iBw1 
     jja = jja + iBant 
     ja = jja 
     ELSE 
      jug = I 
      ja = j 
     END IF 
     FOR k = jug TO jog 
      s = s - a(ja) * b(k) 
      ja = ja + iBw1 
     NEXT k 
     b(ij) = s / a(ia) 
     ia = ia + iBant 
    NEXT j 
    ia = nibw - iBw1 
    ii = ii + n 
    b(ii) = b(ii) / a(ia) 
    FOR j = 1 TO n1 
     ij = ii - j 
     ia = ia - iBant 
     s = b(ij) 
     ja = ia + 1 
     jug = ij + 1 
     jog = ii 
     IF j > iBw1 THEN jog = ij + iBw1 
     FOR k = jug TO jog 
      s = s - a(ja) * b(k) 
      ja = ja + 1 
     NEXT k 
     b(ij) = s / a(ia) 
    NEXT j 
   NEXT I 
 
END SUB 'AsagiYukariHesap 

 

Bu alt program CholeskyBant1 alt 
programının hemen arkasından çağrılır,  
Aşağı ve yukarı doğru hesap ile bilinmeyen 
matrisi x i hesaplayarak b nin üzerine 
depolar. 


